

T.P. 2 – Corrigé

Space Invaders (partie 5)

Étape 1

```
ClearScreen      ; Sauvegarde les registres.
                 move.l  d0,-(a7)

                 ; Remplit la mémoire vidéo avec des zéros.
                 moveq.l #0,d0
                 jsr     FillScreen

                 ; Restaure les registres puis sortie.
                 move.l  (a7)+,d0
                 rts
```

Étape 2

```
BufferToScreen   ; Sauvegarde les registres.
                 movem.l d7/a0/a1,-(a7)

                 ; Tampon vidéo -> A0.L
                 lea     VIDEO_BUFFER,a0

                 ; mémoire vidéo -> A1.L
                 lea     VIDEO_START,a1

                 ; Nombre d'itérations - 1 -> D7.W (car DBRA)
                 ; Nombre d'itérations = Nombre de mots longs
                 move.w  #VIDEO_SIZE/4-1,d7

\loop            ; Copie un mot long du tampon vers la mémoire vidéo.
                 move.l  (a0),(a1)+

                 ; Met à 0 le mot long du tampon qui vient d'être copié.
                 clr.l   (a0)+

                 ; Reboucle tant qu'il y a des mots longs à copier.
                 dbra    d7,\loop

                 ; Restaure les registres puis sortie.
                 movem.l (a7)+,d7/a0/a1
                 rts
```

Étape 3

```
PrintSprite      ; Sauvegarde les registres.
                 movem.l d1/d2/a0,-(a7)

                 ; Si le sprite ne doit pas être affiché, on quitte.
                 cmp.w   #HIDE,STATE(a1)
                 beq     \quit

                 ; Affiche le sprite.
                 move.w  X(a1),d1
                 move.w  Y(a1),d2
                 movea.l BITMAP1(a1),a0
                 jsr     PrintBitmap

\quit            ; Restaure les registres puis sortie.
                 movem.l (a7)+,d1/d2/a0
                 rts
```