

T.P. 5 – Corrigé

Space Invaders (partie 8)

Étape 1

```
PrintShip      ; Sauvegarde les registres.
               move.l  a1,-(a7)

               ; Affiche le vaisseau.
               lea     Ship,a1
               jsr     PrintSprite

               ; Restaure les registres puis sortie.
               move.l  (a7)+,a1
               rts
```

```
MoveShip      ; Sauvegarde les registres.
               movem.l d1/d2/a1,-(a7)

               ; Initialise le mouvement relatif à zéro.
               clr.w   d1
               clr.w   d2

\right        ; Si la touche "droite" est pressée,
               ; incrémente D1.W (déplacement vers la droite).
               tst.b   RIGHT_KEY
               beq     \left
               add.w   #SHIP_STEP,d1

\left         ; Si la touche "gauche" est pressée,
               ; décrémente D1.W (déplacement vers la gauche).
               tst.b   LEFT_KEY
               beq     \next
               sub.w   #SHIP_STEP,d1

\next         ; Déplace le vaisseau (en fonction de D1.W et de D2.W).
               lea     Ship,a1
               jsr     MoveSprite

               ; Restaure les registres puis sortie.
               movem.l (a7)+,d1/d2/a1
               rts
```

Étape 2

```
PrintShipShot      ; Sauvegarde les registres.
                   move.l  a1,-(a7)

                   ; Affiche le tir du vaisseau.
                   lea     ShipShot,a1
                   jsr     PrintSprite

                   ; Restaure les registres puis sortie.
                   move.l  (a7)+,a1
                   rts
```

```
MoveShipShot      ; Sauvegarde les registres.
                   movem.l a1/d1/d2,-(a7)

                   ; Adresse du sprite de tir -> A1.L
                   lea     ShipShot,a1

                   ; Si le tir n'est pas affiché, on ne fait rien.
                   cmp.w   #HIDE,STATE(a1)
                   beq     \quit

                   ; Déplace le tir vers le haut.
                   ; Si le déplacement a eu lieu, on peut quitter.
                   clr.w   d1
                   move.w   #-SHIP_SHOT_STEP,d2
                   jsr     MoveSprite
                   beq     \quit

\outOfScreen      ; Sinon, le tir sort de l'écran.
                   ; Il faut alors le camoufler.
                   move.w   #HIDE,STATE(a1)

\quit             ; Restaure les registres puis sortie.
                   movem.l (a7)+,a1/d1/d2
                   rts
```

```

NewShipShot      ; Sauvegarde les registres.
                  movem.l d1-d3/a0/a1,-(a7)

                  ; Si la touche espace est relâchée, on ne fait rien.
                  tst.b  SPACE_KEY
                  beq     \quit

                  ; Si le tir de vaisseau est déjà présent, on ne fait rien.
                  lea     ShipShot,a0
                  cmp.w   #SHOW,STATE(a0)
                  beq     \quit

                  ; Coordonées du vaisseau -> Coordonées du tir
                  lea     Ship,a1
                  move.w  X(a1),X(a0)
                  move.w  Y(a1),Y(a0)

                  ; Largeur du vaisseau -> D1.W
                  movea.l BITMAP1(a1),a1
                  move.w  WIDTH(a1),d1

                  ; Hauteur du tir -> D2.W
                  ; Largeur du tir -> D3.W
                  movea.l BITMAP1(a0),a1
                  move.w  HEIGHT(a1),d2
                  move.w  WIDTH(a1),d3

                  ; Centre le tir horizontalement par rapport au vaisseau.
                  sub.w   d3,d1
                  lsr.w   #1,d1
                  add.w   d1,X(a0)

                  ; Positionne le tir juste au dessus du vaisseau.
                  sub.w   d2,Y(a0)

                  ; Le tir est rendu visible.
                  move.w  #SHOW,STATE(a0)

\quit            ; Restaure les registres puis sortie.
                  movem.l (a7)+,d1-d3/a0/a1
                  rts

```